



THREAT HUNTING GUIDE

How to threat hunt with Open NDR + MITRE ATT&CK®



This Threat Hunting Guide was created to teach you simple and relevant ways to discover attacks before they happen using Corelight network data. This document—organized around the MITRE ATT&CK® framework—is designed to help you develop a theory for threat hunting and establish prioritization.

MITRE ATT&CK is a globally-accessible knowledge base of adversary tactics and techniques based on real-world observations. It's used as a foundation for specific threat models and methodologies in the private sector, government, and the cybersecurity industry. With the creation of ATT&CK, MITRE is fulfilling its mission to solve problems for a safer world—by bringing communities together to develop more effective cybersecurity. ATT&CK is open and available to any person or organization for use at no charge.¹

WHAT IS THREAT HUNTING?

At a high level, threat hunting is actively looking for adversaries in your network when you don't know if they're inside. This is different from indicator matching, which is only watching for well-known signs of attackers, for example, IP addresses or file hashes. Usually conducting a threat hunt involves researching a theory, or hunch, and then analyzing data looking for something interesting. Items that are interesting can take many shapes, for example in *The Cuckoo's Egg* by Clifford Stoll, an accounting error initiated the hunt.

"Dave wandered into my office, mumbling about a hiccup in the Unix accounting system. Someone must have used a few seconds of computing time without paying for it. The computer's books didn't quite balance; last month's bills of \$2,387 showed a 75-cent shortfall."

This 75-cent difference was the indicator that led to the discovery of the compromise of multiple corporations and government systems. The term "interesting" is used throughout this guide and it is only limited by your imagination.

WHY CONDUCT A THREAT HUNT?

Most host- or network-based detection systems rely on matching, otherwise known as signatures, to generate alerts to signal defenders that there is something unwanted in

the network. However, attackers are continually evolving to evade detection, and signatures are developed only after the artifact was discovered in another network. So, if you're not hunting for artifacts in your environment, how will you discover that attackers are evading your current defenses?

Hunting has several positive outcomes. The first is you might find artifacts of an active intruder that your current defenses missed. While some may think this is a tragedy, it can be a huge win, especially if the intruder hasn't completed their objective(s). In every hunt, there's always something to find.

You may discover network or software misconfigurations that pose a threat, either because they degrade network performance or introduce a vulnerability. Next, the hunt could yield run-of-the-mill infections such as adware, or other dormant malware that aren't directly targeting your organization but are still a threat. Lastly, resource abuse and Shadow IT, services that are not officially supported, can introduce risk through degraded network performance or new adversary attack vectors. Every hunt teaches you something new about the network which will aid in your next investigation.

WHY HUNT WITH NETWORK DATA?

PACKETS. DON'T. LIE.

It's really as simple as that. If a network-resident intruder is active in your network, there will be network artifacts. In artifacts, there are clues to what is happening, or better yet, an exact moment-for-moment story of what happened. For example, if a command and control channel uses DNS as a transport mechanism, there will be DNS queries and replies. Additionally, the IP address(es) that are on the ends of a TCP connection must be accurate, they cannot be spoofed if data is exchanged. All attacks traverse the network, unless they are isolated to one host, so there will be packets.

CORELIGHT LOGS NOMENCLATURE

Corelight provides data-centric solutions that analyze network traffic and enhance automation tools by transforming network traffic into linked logs and extracting files. The central log is the **conn** log, which documents general information about all network sessions.

The **conn** log records information about each network endpoint and the **service** (application) and also assigns a **uid** (unique identifier). The **uid** links the **conn** log to related

protocol logs, where specific session information is available. For example, the **conn** log can list **http** as the **service**, and using the **uid** you can pivot to the **http** log to get specific protocol information about the session. The **uid** separates Corelight solutions from other security tools. This field links otherwise disparate information into easily digestible logs. The **uid** is fundamental to conducting link analysis and a critically important field that facilitates pivoting, or joining multiple logs together.




Table JSON

Single Document bro@tguicGE8U/cpRQ0oDhb8

Table JSON

Single Document bro@phrteGE8U/cpRQ0ov47G

Table JSON

The information about each network endpoint is summarized by the **id** field, which is usually represented as four separate fields:

- **id.orig_h**
- **id.orig_p**
- **id.resp_h**
- **id.resp_p**

This nomenclature may seem odd to use, because networking personnel traditionally refer to sessions using client and server; however, using **orig** (originator) and **resp** (responder) allows security personnel to accurately describe the connection. Think of the originating host (**orig_h**) as the source, or client, and the responding host (**resp_h**) as the destination, or server. The fields **id.orig_p** and **id.resp_p** will be populated with the corresponding port numbers.

Many of the remaining fields within the **conn** log and other protocol logs are self-descriptive, but if you get stuck, look at the Zeek® documentation at <https://docs.zeek.org/en/current/> for more detailed information or visit the Zeek community Slack channel at <https://zeek.org/slack>.

IDENTIFYING USERS AND DEVICES

When identifying devices on a network, the IP or MAC addresses are regularly used to create the 'identity.' The device IP address is used more often for the remote identity of a device because it survives router boundaries. When inside a network segment, the MAC address is preferred for identification because it can be a reliable identifier of a specific machine. Each identifier has pros and cons, and the ability of Corelight to capture both aids SOC personnel as they investigate events.

While IP addresses are durable² for internal investigations, they often are transient within a network due to most networks implementing DHCP (Dynamic Host Configuration Protocol). Transient IPs are problematic for defenders when the IDS alert identifies the session by IP addresses. Those IP addresses are only related to the alert at the time that the alert was generated.

You can use open source tools when conducting an investigation (e.g., **nslookup**), to provide DNS information for remote IPs. However, this is a point-in-time piece of information at the time of the investigation, not when the event occurred. A better technique is to use logs created at the time of the alert to capture the IP and FQDN (fully qualified domain name) for the remote device. To locate internal devices, you can mine DHCP logs to identify them. There are multiple ways to identify a host and Corelight provides this data in multiple logs that each tell a different aspect of the story. Exercise creativity and follow every lead.

Where hostnames can be found:

- **known_names**: with the Known Entities package enabled, Corelight sensors collect many of the following sources of information into one log updated every fifteen minutes. In the **known_names** log, the **hostname** field refers to the device name obtained from the network traffic, the **host_ip** field refers to the IP address that the name is attributed to, and the **protocols** array represents what protocol(s) the hostname was observed from.
- **dhcp**: the **host_name** and **domain** fields represent the hostname and domain reported by a host when requesting an IP address via DHCP, and the **assigned_addr** field is the IP address that was assigned to that host.
- **dns**: if there's an IP in the **answers** field, then the **query** field contains the hostname that the DNS server recorded (at that time) for the IP address.
- **ntlm**: **server_dns_computer_name** and **server_nb_computer_name** refer to the DNS and Netbios names of the machine with the IP address in the **id.resp_h** field. The **hostname** field is the hostname of the machine with the IP address in the **id.orig_h** field.
- **kerberos**: in a Windows environment, for domain-joined devices, Kerberos requests where the **client** field contains a name ending in \$, the **client** field is the hostname, and the **id.orig_h** field is the IP address of that host. The **client** field is often structured like **HOSTNAME\$EXAMPLEDOMAIN.COM** where **HOSTNAME** is the hostname and **EXAMPLEDOMAIN.COM** is the Windows domain name and Kerberos realm name.
- **http**: the **host** field contains the hostname, domain name, or IP address of the HTTP server. Sometimes this field is an indication of the identity of the server, the device with the IP address in the **id.resp_h** field, but since this value is asserted by the client there can be exceptions.
- **ssl**: **server_name** field is extracted from the Server Name Indication (SNI) field in the TLS/SSL negotiation, and is used similarly to the **host** field of the **http** log. Also, the **subject** field is extracted from the subject of the server certificate, and the canonical name CN portion of the subject can provide clues to identify a server.

When identifying users, there are several logs that provide valuable information:

- **known_users**: with the Known Entities package enabled, Corelight sensors collect many of the following sources of information into one log updated every fifteen minutes. In the **known_users** log, the **host_ip** field refers to the origin host for authentication protocols (e.g. NTLM, Kerberos) and the responding host for remote access protocols (e.g. RDP), the **remote_ip** field refers to the IP address at the other end of the connection, and the **user** field houses the user name.
- **rdp**: depending on the version of the RDP protocol, the value of the **cookie** field is the username asserted by the client, and the client IP is in the **id.orig_h** field.³

- **ftp**: the **user** field contains the username asserted by the client, and the client IP address will be in the **id.orig_h** field.
- **irc**: the **user** field contains the username asserted by the client, and the client IP address will be in the **id.orig_h** field.
- **socks**: the **user** field contains the username asserted by the client, and the client IP address will be in the **id.orig_h** field.
- **http**: the **username** field contains the username asserted by the client, and the client IP address will be in the **id.orig_h** field, or may be indicated in the **proxied** field if the connection was proxied. If proxied the **id.orig_h** field will contain the IP address of the proxy.
- **ntlm**: the **username** field contains the username asserted by the client, and the client IP address will be in the **id.orig_h** field.
- **kerberos**: in a Windows environment, kerberos requests contain the username in the **client** field (except for requests where the **client** field contains a name ending in \$, which means that the asserting identity is a device, and the **id.orig_h** field is the IP address of the source device. The **client** field will often be structured like **USERNAME/EXAMPLEDOMAIN.COM** where **USERNAME** is the username and **EXAMPLEDOMAIN.COM** is the Windows domain name and Kerberos realm name

A few words of warning about drawing conclusions about the identity of a machine or the user of a device: know your limits (and the limits of the data). Just because a username was recorded in network traffic does not mean that the actual person with that name is responsible—it is just a clue. You should check to see if the user authenticated successfully, as state-sponsored cyberspies and saboteurs have increasingly experimented with planting false flags.⁴ The username could have been *asserted*, but if the authentication failed, then it is not a clear indicator that the user was involved. Don't forget that devices and software may cache credentials, so the user account may be active, but the actual person could still be innocent. You must continue to collect information before you can confirm nefarious behavior.

For example:

- A user goes to lunch and leaves their device unlocked, and someone else uses their machine while the original user is away
- A device is compromised with a Remote Access Trojan (RAT) and a user halfway around the world is surreptitiously assuming the identity of our victim, while the original user is also using the device simultaneously to conduct regular business
- A malicious user within the organization has overheard a coworker saying their password out loud in conversation, and he or she is now trying to use those credentials to log in to other systems

Also, make sure you understand what pieces of information are controlled and asserted by the clients or servers, and consider who controls each. If an adversary is inside your network, determining what information is trustworthy is paramount when preparing the response plan. For example, an intruder could disable DHCP and statically assign an IP address and use it to navigate the network, making identification difficult, as the DHCP server records would provide conflicting information, or no information at all. Additionally, when a client requests a DHCP address, an intruder could provide a false MAC address and/or hostname. Passively capturing point-in-time logs of events will give you your best shot at deciphering what occurred.

INCLUDED QUERIES

This guide includes sample queries to jump-start your threat hunting exercises. They are written for Corelight Investigator using the LogScale Query language (LQL), and should be usable in Investigator, any LogScale instance, or any LQL-compatible system that has Corelight data in it. To try a query out, simply copy it, paste and go!

INDEX OF TTPS

INITIAL ACCESS

Drive-By Compromise	7
External Remote Services	8
Spearphishing Attachment.....	12
Spearphishing Link.....	15

EXECUTION

Command Line Interface, PowerShell.....	16
---	----

PERSISTENCE

BITS Jobs	17
External Remote Services	19
Port Knocking.....	19
Server Software Component: Web Shell	19

DEFENSE EVASION

BITS Jobs	22
Port Knocking.....	22
Install Root Certificate	22

CREDENTIAL ACCESS

Brute Force.....	23
Forced Authentication.....	25
Network Sniffing.....	28

DISCOVERY

Network Service Discovery.....	28
Network Share Discovery	30
Network Sniffing.....	30
Remote System Discovery	30

LATERAL MOVEMENT

Remote Desktop Protocol.....	30
Exploitation of Remote Services.....	30
Windows Admin Shares.....	34

COLLECTION

Archive Collected Data	35
Automated Collection.....	37
Data from Network Shared Drive.....	37

COMMAND AND CONTROL

Data Obfuscation: Protocol or Service Impersonation, Non-Standard Ports	39
Encrypted Channel.....	40
Fallback Channels, Multi-Stage Channels	40
Ingress Tool Transfer	41
Non-Application Layer Protocol.....	44
Proxy	45
Web Service.....	47

EXFILTRATION

Automated Exfiltration.....	48
Data Transfer Size Limits.....	48

HOW TO HUNT FOR SPECIFIC TTPs

INITIAL ACCESS (TA0001)

Initial access is when intruders establish their initial foothold.

Drive-By Compromise (T1189)

A drive-by compromise usually results when a file is surreptitiously downloaded from a website that is compromised. When you hunt for signs of drive-by compromise in Corelight data, your main focus is downloads from external websites. Begin the hunt with the **http** log and look for signs of downloaded executables:

1. Start with **http** logs where **resp_fuids** is not empty. This means there was a file returned from the responder.
2. If the data volume is too large, filter out local (in-network) responders. You can filter by joining the results to the **conn** log on the **uid**, then filtering out any records in which **local_resp** is **true** in the **conn** log.
3. Review the **resp_mime_types** from the **http** log, and filter uninteresting results (e.g., images, text, OCSP responses, and certificates). Often the most interesting results are executables, DLLs, and archives/containers
4. Group the results by the **host** and **resp_mime_types** fields for easy analysis.

Scan through the results and look for anything interesting or odd, such as downloads of executable files, or file extension and mime-type mismatch.

As more attackers move to using TLS to encrypt exchanges between compromised clients and websites they control, there will be less visibility via the **http** log. To regain this visibility, consider using an enterprise SSL decryption solution and passing the decrypted HTTP traffic to your Corelight Sensor.

EXECUTABLE DOWNLOAD DIRECTLY FROM IP

```
(#path="http" or #path="http_red")
| host = /[0-9]{1,3}$
| in(uri, values=[*.apm", "*.app", "*appref-ms", "*bas", "*bat", "*chi", "*chm", "*chq",
  "*chw", "*dll", "*exe", "*gadget", "*hta", "*inf", "*jar", "*jnlp", "*jse", "*lnk",
  "*mde", "*mht", "*msi", "*msix", "*msixbundle", "*pif", "*pkg", "*pl", "*ps1",
  "*ps1xml", "*ps2", "*ps2xml", "*psc1", "*psc2", "*psd1", "*psd1", "*psdm1", "*psm1",
  "*py", "*pyc", "*pyo", "*pyw", "*pyz", "*reg", "*scr", "*sct", "*vbe", "*vbs", "*ws",
  "*wsb", "*wsc", "*wsf", "*xpi", "*xpi", "*xz", "*z", "*zip", "*zipx"])
| split(resp_mime_types)
| groupby([id.orig_h, id.resp_h, host, method, uri, status_code], function=collect(
  resp_mime_types, limit=5))
```

POSSIBLE WINDOWS EXECUTABLE DOWNLOAD WITHOUT MATCHING MIME TYPE

```
(#path="http" or #path="http_red")
| ! (uri="*.exe" or uri="*.dll" or uri="*.msi")
| split(resp_mime_types)
| in(resp_mime_types, values=[
"application/java-archive",
"application/mshelp",
"application/chrome-ext",
"application/x-object",
"application/x-executable",
"application/x-dosexec",
"application/x-msdownload",
"application/vnd.microsoft.portable-executable"])
| groupby([id.orig_h, id.resp_h, id.resp_p, host, method, uri, status_code],
function=collect(resp_mime_types, limit=5))
```

DOWNLOAD OF FILE FROM INTERNET (OVERVIEW QUERY)

```
definetable({{#path=conn or #path=conn_red) local_resp=false local_orig=true
service="*http*"}, name=outbound_conns, include=uid)
| (#path=http or #path=http_red) status_code=200 resp_mime_types[0]=*
| match(file=outbound_conns, field=uid)
| split(resp_mime_types)
| !in(field=resp_mime_types, values=["text/*", "image/*", "application/xml",
"application/font", "video/*", "application/ocsp-response", "application/pgp-signature",
"application/pdf", "application/vnd.ms-opentype", "application/x-font"])
| groupby([id.orig_h, id.resp_h, id.resp_p, host, method, uri, status_code],
function=collect(resp_mime_types, limit=5))
```

External Remote Services (T1133)

External remote services are used by adversaries to connect to internal network resources, and hunting for misuse of remote services usually involves two steps: discovery, and analysis. First, you must discover what remote services are in use. Asset and service inventory information should be collected first, but usually it's insufficient. Often, there is natural "drift" as IT teams make changes to infrastructure and struggle to keep asset documentation current. Empowered users make this more difficult by setting up assets and services without involving or informing IT, a process known as "shadow IT."

Traditional remote services, for example: RDP, VNC (remote framebuffer), and SSH (secure shell) contain a server component and a client component. If you have a remote service hosted in your environment, attackers can exploit externally accessible services to compromise machines inside the network. To identify these services, look for **conn** log entries in which the **service** field contains **rfb**, **rdp**, or **ssh**, and where **local_orig** is **false** and **local_resp** is **true**, or where the originator IP (**id.orig_h**) is external and the responder IP (**id.resp_h**) is on the organization network. Make note of any RFB/VNC, RDP, or SSH servers that are accepting connections from the internet.

Some remote services work in reverse, where an agent is installed on the local device, and it reaches outward from inside the network to a set of external servers, for example, GoToMyPC and TeamViewer. This configuration is designed to assist users (primarily home users) who don't control the NAT or the firewall or aren't sophisticated enough to be able to manage port forwarding or firewall rule management.

To discover if these remote services are in use in your environment, look for signs of outbound connections to the services. For example, TeamViewer uses TCP port 5938 to communicate with TeamViewer servers, so simply review the **conn** logs for connections where the **id.resp_p** is **5938** and **local_orig** is **true** and **local_resp** is **false**. TeamViewer also uses SSL, and the domain name of the connections should be ***.teamviewer.com**, so additionally you can look for entries in the **ssl** log in which the **server_name** contains, or better yet ends with, **teamviewer.com**. (Note: because this session works in reverse, the **id.orig_h** is the device in your network that has the TeamViewer client installed.) Our second example, GoToMyPC, attempts to contact **poll.gotomypc.com**. Examine the **http** log **host** field for **poll.gotomypc.com**, or entries in the **ssl** log in which the **server_name** is **poll.gotomypc.com**. For each client software package, the list of ports and domain names varies.

Now that we've discussed the discovery of remote services, you should compare Corelight data to a list of all remote services that the IT department offers, such as:

- RDP Gateways
- VDI (Virtual Desktop Infrastructure) Gateways
- VPN (Virtual Private Network) Gateways
- SSH Servers

For each service exposed to the internet, aggregate a list of connections to that service from the **conn** log, and include the following fields:

- **id.orig_h**: Origin IP address (client)
- **id.resp_h**: Responder IP address (server)
- **id.resp_p**: Responder port
- **service**: the application protocol that Corelight detected
- **history**: the history of the connection, e.g. what types of TCP flags were seen
- **orig_cc**: The originator's country code

When filtering logs, ensure the **history** field starts with **Sh**. For TCP connections this means that the originator sent a SYN, and the responder replied with a SYNACK (handshake). This check eliminates connections where the server is not listening, or there is a firewall blocking the connection.

After you have gathered all the data, begin sifting through the logs for anything interesting, such as a connection from a country that is not expected. Use the UID from the **conn** log to follow-up with the application-specific Corelight logs (**rdp**, **rfb**, **ssh**). For example, the **rdp** log contains more details about the connection, such as the **cookie** field that can contain the username of the authenticating user. The last step is to check with the user to determine whether they were actively using the system at that time.

Corelight customers have access to the [Encrypted Traffic Collection \(ETC\)](#) that generates inferences, or insights, about encrypted traffic. The **ssh** log contains interesting information inferred about the SSH connection, such as:

- **KS** for connections that appear to contain client keystrokes
- **FU** and **FD** for connections which appear to contain a file upload or download, respectively
- **ABP** for connections which appear not to contain any authentication, but still are successful ("authentication bypass")
- **SV** or **SC** for clients that appear to be version or capability scanning, respectively

If you'd like to learn more about the Corelight ETC, please contact our sales team at (510) 281-0760.

RDP SCANNING POTENTIAL BRUTE FORCE COMMON USER NAMES

```
#path="rdp" | lowercase(cookie)
| in(field=cookie, values=["root", "administr", "admin", "guest", "info", "test", "adm",
"user", "da", "local", "letmein", "service", ".", "computer", "xxx", "/", "\\"})
| groupby(field=id.orig_h, function=[collect(cookie, limit=10), count(field=cookie,
distinct=true, as=val)])
| val > 2
```

RESPONSE FROM EXTERNAL FACING SERVICE (OVERVIEW QUERY)

```
(#path="conn" or #path="conn_red") local_orig="false" local_resp="true" history="Sh*"
| splitString(field=service, by=",") | split(service)
| groupby(id.resp_h, function=[collect(service, limit=10), count(id.orig_h, distinct=true,
as="#_clients"), count()])
```

EXTERNAL FACING ICS MODBUS

```
(#path=conn or #path=conn_red) service="*modbus*" local_orig=false local_resp=true
| groupby([service, id.resp_h, id.resp_p], function=[count(id.orig_h, distinct=true,
as="#_clients"), count()])
```

EXTERNAL FACING ICS DNP3

```
(#path=conn or #path=conn_red) service="*dnp3*" local_orig=false local_resp=true
| groupby([id.resp_h, id.resp_p], function=[collect(service, limit=10), count(id.orig_h,
distinct=true, as="#_clients"), count()])
```

UNCOMMON EXTERNAL FACING APPLICATION SERVICE

```
(#path="conn" or #path="conn_red") history="Sh*"
| local_orig="false" local_resp="true"
| in(field=service, values=["*dce_rpc*", "*dnp3*", "*gssapi*", "*krb_tcp*", "*krb_udp*",
"*krb*", "*modbus*", "*ntlm*", "*radius*", "*rdp*", "*rdpeudp*", "*rpc*", "*smb*", "*snmp*",
"*syslog*"])
| splitString(field=service, by=",", as="service") | split(service)
| groupby([id.resp_h, id.resp_p], function=[collect(service, limit=10), count(id.orig_h,
distinct=true, as="#_clients"), count()])
```

INBOUND RDP FROM INTERNET

```
(#path="conn" or #path="conn_red") history="Sh*"
local_orig=false local_resp=true service="*rdp*"
| splitString(field=service, by=",", as=service) | split(service)
| groupby([id.resp_h, id.resp_p], function=[collect(service, limit=10), count(id.orig_h,
distinct=true, as="#_clients"), count()])
```

INBOUND SSH FROM INTERNET

```
(#path="conn" or #path="conn_red") history="Sh*"
local_orig=false local_resp=true service="*ssh*"
| splitString(field=service, by=",", as=service) | split(service)
| groupby([id.resp_h, id.resp_p], function=[collect(service, limit=10), count(id.orig_h,
distinct=true, as="#_clients"), count()])
```

OUTBOUND CONNECTION ON KNOWN TEAMVIEWER PORT 5938

```
(#path="conn" or #path="conn_red") local_orig=true local_resp=false id.resp_p=5938
| groupby([id.orig_h, id.resp_p], function=[collect(history, limit=5), collect(id.resp_h,
limit=5), count()])
```

SSL SNI INVOLVES TEAMVIEWER

```
(#path="ssl" or #path="ssl_red") (server_name=*.teamviewer.com or server_name=teamviewer.com)
| groupby([id.orig_h, id.resp_p], function=[collect(server_name, limit=5), collect(id.resp_h,
limit=5), count()])
```

SSL SNI INVOLVES GOTOMYPC

```
(#path="ssl" or #path="_red") (server_name=*.poll.gotomypc.com or
server_name=poll.gotomypc.com)
| groupby([id.orig_h, id.resp_p], function=[collect(server_name, limit=5), collect(id.resp_h,
limit=5), count()])
```

Spearnishing Attachment (T1566.001)

As a method of entry into an organization, an adversary may send a well-crafted malicious attachment to an individual or a small group in a spearphishing campaign. The attachment could be a document that instructs the user to take some action, such as clicking a link and/or logging in to a portal; or it could be a file crafted to exploit a vulnerability in the software used to open it, such as Adobe Acrobat or Microsoft Word. The Corelight **smtp** log contains records in the **uids** field if there were any files attached to a message delivered over SMTP. This field can be used to pivot to the files log which contains detailed information about the file including filename, hashes, and the source.

For example, examine this sample log below.

```
path: smtp
from: Your Friend <Jeremy.Rigeur@gmail.com>
uids: [ "Fh5GBc1wdVp3x9MKxc" ]
mailfrom: attacker@fake-mail.com
rcptto: [ "victim@corp-mail.com" ]
subject: Definitely not a spear-phish
to: [ "victim@corp-mail.com" ]
uid: CzKseq1Y3zo2qsTYH5
user_agent: Apple Mail (2.3608.80.23.2.2)
```

```
path: files
conn_uids: [ "CzKseq1Y3zo2qsTYH5" ]
filename: WIRE_FRAUD.pdf
fuid: Fh5GBc1wdVp3x9MKxc
md5: e71c36cddd2aa42670d89d63e653d1da
mime_type: application/pdf
sha1: bb24829550c0ca17db73d80a1d2f969e3b06ff5f
source: SMTP
```

To hunt for potential spearphish attempts, you can search in the **files** log:

1. The value in the **source** field is **SMTP**.
2. Filter out any uninteresting **mime_type** and/or **filename** values, as previously mentioned.
3. Use the hash (MD5, SHA1, or SHA256) with a file reputation service (such as Virustotal) to look for known malicious files.

Additionally, you may start from the **smtp** log:

1. To reduce the data look for entries where the **uids** field is non-empty.
2. Filter out known good combinations of **mailfrom** and **from** values.
3. Filter out uninteresting **subject** values.
4. Consider using the **fuid** value from the remaining records to pivot to the **files** log to get more information about the file.

Corelight can perform high-speed file extraction and can filter based on MIME type, so any interesting files, such as executables, Office documents, and PDFs are available for more scrutiny if desired.

Much of the mail that crosses the internet today is encrypted via STARTTLS over the SMTP protocol, and this hinders visibility. To achieve better visibility without sacrificing privacy and security for your users, it is a best practice to accept inbound SMTP at a system that supports STARTTLS, then proxy the mail to the internal mail system, so that Corelight can generate the corresponding logs.

POTENTIALLY HARMFUL ATTACHMENT

```
(#path="files" or #path="files_red") source="SMTP"
| in(field=filename, values=[
  "*.7z", "*.ace", "*.apm", "*app", "*appref-ms", "*.arj",
  "*.asp", "*bas", "*bat", "*bz2", "*bzip2", "*cab", "*cdxml",
  "*cer", "*chi", "*chm", "*chq", "*chw", "*class", "*cmd",
  "*cnt", "*com", "*cpl", "*crt", "*doc", "*docm", "*epub",
  "*exe", "*gadget", "*gz", "*gzip", "*hta", "*img", "*inf",
  "*ins", "*ins", "*iso", "*isp", "*isp", "*jar", "*jar",
  "*jnlp", "*jse", "*lnk", "*lzh", "*mde", "*mht", "*msi",
  "*msix", "*msixbundle", "*ods", "*odt", "*pif", "*pkg",
  "*pl", "*ps1", "*ps1xml", "*ps2", "*ps2xml", "*psc1",
  "*psc2", "*psd1", "*psd1", "*psdm1", "*psm1", "*pssc",
  "*py", "*pyc", "*pyo", "*pyw", "*pyz", "*pyzw", "*r01",
  "*r14", "*r18", "*r25", "*rar", "*reg", "*scr", "*sct",
  "*shb", "*sys", "*tar", "*taz", "*tbz", "*tbz2", "*tgz",
  "*txz", "*udl", "*vbe", "*vbs", "*ws", "*wsb", "*wsc",
  "*wsf", "*xbap", "*xls", "*xlsm", "*xpi", "*xz", "*z",
  "*zipx"])
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, mime_type, filename]),
limit=10]))
```

POSSIBLE TYPO SQUATTING OR PUNYCODE PHISHING HTTP REQUEST

```
(#path="http" or #path="http_red") method="GET" !(referrer="*") host="*xn--*"
| groupby([id.resp_h, id.resp_p], function=[collect([id.orig_h, method, status_code, host,
uri, user_agent], limit=10), count()]))
```

FILES AS SMTP ATTACHMENTS (OVERVIEW QUERY)

```
(#path="files" or #path="files_red") source="SMTP"
| !in(mime_type, values=[
  "text/*", "image/*", "application/xml", "application/font",
  "video/*", "application/ocsp-response", "application/pgp-signature",
  "application/vnd.ms-opentype", "application/x-font"
])
| groupby([id.orig_h, id.resp_h, id.resp_p, mime_type, filename, md5])
```

Spearphishing Link (T1566.002)

Instead of sending files into an organization where they can be scrutinized by a corporate mail filter, some adversaries send emails that only contain links. These links lead to websites that are controlled by the attacker, and attempt to dupe the user into:

- Entering credentials that the attackers harvest
- Exploiting a vulnerability in the user's browser
- Downloading a file to exploit another application on the user's device

Corelight Sensors have a package⁵ that can log links from SMTP messages into a separate log, the **smtp_links** log. This log contains a **fuid** field, which links the **smtp_links** log to the **smtp** log. You can quickly pivot to the **smtp** log with the details about the message that delivered the malicious link. For example, see the log examples.

```
path: smtp_links
fuid: FhahXA1eJ32gHvNP27
id.orig_h: 172.16.0.10
id.orig_p: 62345
id.resp_h: 10.0.1.10
id.resp_p: 25,
link: http://www.hamsterwaffle.com/dl.php?id=jimmydean37
uid: C62tx01FH0JFJpsgP1
```

```
path: smtp
from: Your Friend <Jeremy.Rigeur@gmail.com>
fuids: [ "FhahXA1eJ32gHvNP27" ]
mailfrom: attacker@fake-mail.com
rcptto: [ "victim@corp-mail.com" ]
subject: Click this link, please
to: [ "victim@corp-mail.com" ]
uid: C62tx01FH0JFJpsgP1
user_agent: Apple Mail (2.3608.80.23.2.2)
```

To hunt for spearphishing links, start with the **smtp_links** log and review the **link** field, filtering out benign domains until you find interesting results. Another option is to join the **smtp_links** log to the **smtp** log via the **fuids** or **uid** field, and filter out benign combinations of **mailfrom** and **from** fields to look for messages from unique senders.

Much of the mail that crosses the internet today is encrypted via STARTTLS over SMTP. To achieve better visibility without sacrificing privacy and security for your users, it is a best practice to accept inbound SMTP at a system that supports STARTTLS, then proxy the mail to the internal mail system, so that a Corelight solution can generate the corresponding logs.

SMTP EMAIL CONTAINING NON ASCII CHARACTERS WITHIN THE SUBJECT

```
(#path="smtp" subject=/^.*[^\x00-\x7F].*$/
```

MULTIPLE CLIENTS TO HTTP USING UNICODE HOST VIA HTTP - POSSIBLE MULTIPLE PHISHING ATTEMPTS

```
(#path="http" or #path="http_red") (method="POST" or method="PUT")
| !(referrer!="") !(uri="/") host="*xn--*"
| groupby(field=id.orig_h, function=[collect(host, limit=5), count(field=uri, distinct=true, as=val)])
| val >10
```

LINKS IN SMTP MESSAGES (OVERVIEW QUERY)

```
#path=smtp_links
| groupby([id.orig_h, domain], function=[collect(link, limit=10), count()])
```

EXECUTION (TA0002)

The adversary is trying to run malicious code.

Command and Scripting Interpreter: PowerShell (T1059.001)

Command line interface scripting has long been used to manage *nix-based systems, and the ability to build and execute scripts is often exploited by attackers. For years there was no equivalent available on Windows, and in the early 2000s Microsoft began development of a new approach to command line management. Soon thereafter, PowerShell (PS) 1.0 was created. PS, in its various iterations, is a built-in tool based on the .NET framework that's used to automate system administration tasks. It provides an interface for users to access services of the Windows operating system.

Although certain PS commands are restricted by default, many commands are available to obtain system information without an executable file. Adversaries use LNK files to bypass safeguards and execute a PS script. LNK files are usually seen as shortcuts, generally found on users' Desktop and Start Menu.

Malicious LNK files are often embedded within what appears to be legitimate documents or pictures. Once opened, the LNK executes a legitimate windows application CMD.exe or MSHTA.exe to bypass security settings. Corelight's file extraction capabilities and integration with various intel platforms provide insight into malware obfuscated by file type. By utilizing Corelight's built-in filtering, you can tune the file extraction parameters to target specific mime-types that are commonly used for malware delivery, including:

- Compressed files
- Microsoft Office (Word, PowerPoint, etc)
- PDF files
- TXT files (PowerShell, VBS)

LNK FILE DOWNLOAD OR USAGE OVER HTTP

```
(#path="http" or #path="http_red") method="GET" !(referrer="*")
| in(uri, values=["*.lnk", "*.LNK", "*inf", "*INF"])
| groupby([id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(host, limit=10), collect(uri, limit=10), collect(status_code, limit=10)])
```

LNK FILE DOWNLOAD OR USAGE OVER SMB (OVERVIEW QUERY)

```
#path=smb_files
| in(field=name, values=["*.lnk", "*.LNK", "*inf", "*INF"])
| groupby([id.orig_h, id.resp_h, name], function=collect(action, limit=10))
```

PERSISTENCE (TA0003)

Persistence is the adversary trying to maintain their foothold.

BITS Jobs (T1197)

Microsoft Background Intelligent Transfer Service (BITS) was created in 2001 as a mechanism for managing file transfers that minimize disruption to the end user. BITS is commonly used to download Windows updates and other software updates from major vendors.

Attackers have two methods of abusing BITS:

- The most common is to create a BITS transfer job directly on a host, allowing a download of secondary payloads through a built-in Windows service that typically bypasses firewalls and other security controls.
- Another alternative is to exfiltrate data through a BITS upload job. Uploads must connect to an IIS server for BITS to function properly, but this requirement is trivial for malware authors to subvert.

Data transfers using the BITS service can take place over HTTP, SSL, and SMB. When BITS uses HTTP traffic, there is a distinctive User-Agent string of "Microsoft BITS/7.5" (or 7.8 in later versions). Unfortunately, there are no distinguishing characteristics of BITS SSL and SMB network traffic. Therefore, the presence of BITS network traffic is not necessarily suspicious, because it is present anywhere Windows machines are connected to the internet. Analysts still can use Corelight data to assess if the BITS traffic is legitimate by analyzing remote systems being used for BITS data transfers. If they are outside of CDNs or major software providers' networks, all BITS uploads should be investigated until proven benign, as this use case is especially rare among legitimate software vendors.

The code sample below is an **http** log showing what the BITS data looks like if it is over HTTP.

```

_path: http
uid: Ca9LrF3x15kVCxe2K4
id.orig_h: 10.10.199.31
id.orig_p: 49987
id.resp_h: 151.205.0.135
id.resp_p: 80
trans_depth: 1
method: GET
host: 151.205.0.135
uri: /pdata/0731497c8fa1dce5/download.windowsupdate.com/d/msdownload/update/software/
secu/2018/05/windows10.0-kb4103723-x64_0722ab30824410046f954417ada8556d2ac308a6.cab
version: 1.1
user_agent: Microsoft BITS/7.8
request_body_len: 0
response_body_len: 1333068983
status_code: 200
status_msg: OK
resp_fuids: FD283F3hrZH8yzYmb8
resp_filenames: windows10.0-kb4103723-x64_0722ab30824410046f954417ada8556d2ac308a6.cab
resp_mime_types: ["application/vnd.ms-cab-compressed"]
accept_encoding: identity
accept: */*

```

MICROSOFT BITS LEAVING THE NETWORK

```

definetable({#path=conn service=/http/}, include=[id.orig_h, local_orig],
name=orig_locality)
| definetable({#path=conn service=/http/}, include=[id.resp_h, local_resp],
name=resp_locality)
| #path=http user_agent=/microsoft bits/i
| match(file=orig_locality, field=id.orig_h, include=local_orig)
| match(file=resp_locality, field=id.resp_h, include=local_resp)
| local_orig=true local_resp=false
| groupby([id.orig_h, id.resp_h, id.resp_p, host, query, user_agent])

```

External Remote Services

See [Initial Access: External Remote Services](#)

Traffic Signaling: Port Knocking (T1205.001)

Port knocking is a technique to get a remote system to enable access to an otherwise closed port. It typically consists of a pre-defined sequence of connections to other (often closed) ports, sometimes with special protocol-level flags, Layer 7 banner strings, etc.

Corelight summarizes each TCP, UDP, and ICMP connection in the **conn** log. This detailed log provides useful statistics about connections. The **history**, **conn_state**, and network tuple (src/dest ip/port) fields provide the information necessary to sight port knocking. It is important to note that sighting port knocking without an additional hint can be a daunting task, as it is easy to hide intentional sequences of connections among the noise of a typical network.

Server Software Component: Web Shell (T1505.003)

A web shell is a web-based implementation of a command shell. A web shell is generally a malicious web page or code snippet introduced into an existing web server or application to provide unauthorized access. This access can be an actual CLI shell, file management, or database access tool. This is a common tactic used by web shells that includes blending malicious traffic with benign traffic to/from the web server, making it difficult to identify via IDS signatures due to the ease of changing specific web shell characteristics.

When a web shell executes, it runs with limited web server software user permissions. Attackers often use web shells to attempt privilege escalation attacks by exploiting local vulnerabilities on the system to assume root privileges.

Detecting web shells on the network using signature-based detections is relatively straightforward, as web shells have specific file paths, communication methods, or other behaviors that can trigger an alert. However, like most 'atomic' IOCs, they are easy to evade because they identify specific behaviors that can easily be changed. Therefore, it is recommended to supplement signature detection with a threat hunting program to find more general behaviors of anomalous activity.

Web shells attempt to hide malicious activity in normal HTTP traffic, making the **http** log an excellent data source for investigating web shell activity. Examples of hunt hypotheses supported by Corelight HTTP data are:

- Unusual HTTP POST activity. This may be as simple as unexpected HTTP POSTs in the **method** field of the **http** log where GETs are expected (if the affected site is primarily serving content).
- 'Normal' web traffic travels to a shortlist of common pages, with navigation via an internal hyperlink. A web shell goes directly to the hidden page and appears as an HTTP request with no referring page. Additionally, web traffic shows a variety of requesting IPs, user-agent strings, JA3s, etc. A web shell can have a more homogenous group of users.
- Ferreting out suspicious logins originating from internal subnets to DMZ servers and vice versa.

This type of hunt analysis and anomaly detection is an effective way to identify malicious (or suspicious) activity, but modern networks are noisy, chaotic places. As with most hunts, you must know what 'normal' data looks like so that you can successfully filter it out.

For more information about preventing and detecting web shells, visit the NSA's guidance at <https://github.com/nsacyber/Mitigating-Web-Shells>.

POSSIBLE WEB SHELL PUT OR POST TO UNUSUAL EXTENSIONS

```
(#path="http" or #path="http_red") and (method="POST" or method="PUT") status_code="2*"
!(request_body_len="0" or response_body_len="0")
| in(uri, values=[
  "*.jpg", "*.jpeg", "*.gif", "*.png", "*.icon", "*.ico",
  "*.xml", "*.swf", "*.svg", "*ppt", "*pttx", "*doc", "*docx", "*rtf",
  "*pdf", "*tif", "*zip", "*mov"
])
| format("%.30s", field=post_body, as=post_body)
| split(resp_mime_types)
| groupby([id.orig_h, id.resp_h, id.resp_p, host], function=[collect(method, limit=10),
collect(uri, limit=5), collect(post_body, limit=5), collect(resp_mime_types, limit=5),
collect(status_code, limit=5)])
```

POSSIBLE WEB SHELL - RARE PUT OR POST BY IP

```
(#path="http" or #path="http_red") (method="POST" or method="PUT") !(status_code="4*")
| in(uri, values=[
  "*.aspx", "*.aspx?", "*.asp", "*.asp?", "*.php", "*.php?", "*.jsp",
  "*.jsp?", "*.jspx", "*.jspx?", "*.war", "*.war?", "*.ashx", "*.ashx?",
  "*.asmx", "*.asmx?", "*.ascx", "*.ascx?", "*.asx",
  "*.asx?", "*.cshtml", "*.cshtml?", "*.cfm", "*.cfm?",
  "*.cfc", "*.cfc?", "*.cfml", "*.cfml?", "*.wss", "*.wss?",
  "*.do", "*.do?", "*.action", "*.action?", "*.pl", "*.pl?",
  "*.plx", "*.plx?", "*.pm", "*.pm?", "*.xs", "*.xs?", "*.t",
  "*.t?", "*.pod", "*.pod?", "*.php-s", "*.php-s?",
  "*.pht", "*.pht?", "*.phar", "*.phar?", "*.phps",
  "*.phps?", "*.php7", "*.php7?", "*.php5", "*.php5?",
  "*.php4", "*.php4?", "*.php3", "*.php3?", "*phtml", "*phtml?",
  "*.py", "*.py?", "*.rb", "*.rb?", "*rhtml", "*rhtml?",
  "*.cgi", "*cgi?", "*.dll", "*.dll?", "*ayws", "*ayws?",
  "*erb", "*erb?", "*rjs", "*rjs?",
  "*.hta", "*hta?", "*.htc", "*.htc?", "*.cs", "*.cs?", "*kt",
  "*.kt?", "*.lua", "*.lua?", "*vbhtml", "*vbhtml?"
])
| groupby(field=uri, function=[collect(host, limit=5), count(field=id.orig_h, distinct=true,
as="#_clients")])
| "#_clients" < 3
```

POSSIBLE WEBSHELL - DIRTY WORD LIST

```
(#path="http" or #path="http_red") !(status_code="4*") (method="POST" or method="PUT")
| in(uri, values=[
  "*pwned*", "*owned*", "*backdoor*", "*spy*", "*bypass*", "*root*",
  "*r00t*", "*p0wn*", "*robots*", "*hidden*", "*shell*", "*crap*",
  "*telnet*", "*hidden*", "*predator*", "*safe_mode*", "*cfexec*",
  "*botp*", "*zer0*", "*mysql_*", "*oracle_*", "*perlbot*", "*.aspx",
  "*.asp", "*.php", "*.jsp", "*.jspx", "*.war", "*.ashx", "*.asmx", "*.ascx",
  "*.asx", "*.cshtml", "*.cfm", "*.cfc", "*.cfml", "*.wss", "*.do",
  "*.action", "*pl", "*plx", "*pm", "*xs", "*t", "*pod",
  "*php-s", "*pht", "*phar", "*phps", "*php7", "*php5",
  "*php4", "*php3", "*phtml", "*py", "*rb", "*rhtml", "*cgi",
  "*dll", "*ayws", "*cgi", "*erb", "*rjs", "*hta", "*htc",
  "*cs", "*kt", "*lua", "*vbhtml"
])
| format("%.30s", field=post_body, as=post_body)
| groupby(field=[id.orig_h, id.resp_h, id.resp_p, host], function=[collect(uri, limit=10),
  collect(method, limit=10), collect(post_body, limit=5)])

```

POSSIBLE DIRECTORY TRAVERSAL WEB SERVER ATTACK

```
(#path="http" or #path=http_red) !(response_body_len="0" or uri="/")
!(status_code="3*" or status_code="4*")
(post_body=/([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)((\|\%5c|\%255c|\u2216|\%c0%5c|\%c0%80%5c)|(\|\%2f|
%252f|\%25252f|\%c0%af|\%e0%80%af|\%c0%2f|\u2215))([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)|([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)((\|\%5c|\%255c|\u2216|\%c0%5c|\%c0%80%5c)|(\|\%2f|\%2
52f|\%25252f|\%c0%af|\%e0%80%af|\%c0%2f|\u2215))([a-zA-Z\_s]\{4,\})/i
or uri=/([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)((\|\%5c|\%255c|\u2216|\%c0%5c|\%c0%80%5c)|(\|\%2f|
%252f|\%25252f|\%c0%af|\%e0%80%af|\%c0%2f|\u2215))([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)|([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)
([.]\%2e|\%252e|\u002e|\%c0%2e|\%e0%40%ae|\%c0ae)((\|\%5c|\%255c|\u2216|\%c0%5c|\%c0%80%5c)|(\|\%2f|\%2
52f|\%25252f|\%c0%af|\%e0%80%af|\%c0%2f|\u2215))([a-zA-Z\_s]\{4,\})/i
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, method, status_code, uri,
  post_body, user_agent], limit=10), count(uri, distinct=true, as="#_uris"), count()])

```

HTTP POST OR PUT URI NON ASCII CHARACTER

```
#path="http" or #path="http_red"
| (method="POST" or method="PUT") AND uri=/^.*[^\\x00-\\x7F].*$/i
| format("%,.30s", field=post_body, as=post_body)
| groupby([host], function=[collect(id.orig_h, limit=10), collect(method, limit=10),
collect(uri, limit=10), collect(status_code, limit=10), collect(post_body, limit=10)])
```

POSSIBLE WEB SHELL PUT OR POST TO NON-ROOT URI WITHOUT REFERER

```
definetable({{#path=conn or #path=conn_red) local_orig=true}, name=internal_ips,
include=[id.orig_h])
| (#path=http or #path=http_red) status_code>=200 status_code<300 referrer!="" uri!="/"
response_body_len>0
| in(method, values=["PUSH", "POST", "GET"])
| !match(file=internal_ips, field=id.orig_h)
| groupby([host, id.resp_h, id.resp_p], function=[collect(method, limit=5), collect(uri,
limit=15), collect(user_agent, limit=15), collect(id.orig_h, limit=15), collect(status_code,
limit=15), collect(referrer, limit=15)])
```

DEFENSE EVASION

Defense evasion consists of techniques that adversaries use to avoid detection throughout their compromise.

BITS Jobs

See [Persistence: BITS Jobs](#)

Port Knocking

See [Persistence: Port Knocking](#)

Subvert Trust Controls: Install root certificate (T1553.004)

Public certificates are used to establish secure TLS/SSL communications. Root certificates are used to identify the root certificate authority (CA). Root certificates are self-signed and form an anchor of trust for public key cryptography. For example, when a root certificate is installed, the system or application will trust certificates in the root's chain of trust. While no network-level device (e.g., routers and switches) can show the certificate chain installed on a client system, the point of installing a malicious root certificate is to bypass trust validation.

Using Corelight data, you can observe all aspects of the TLS/SSL session using the **ssl** and **x509** logs. These two logs allow analysts to identify certificates that seem suspicious by:

1. Searching the **ssl** log for any entries where the **validation_status** field doesn't have a value of **ok**.
2. Reviewing records where the **validation_status** field is either **self-signed certificate** or **self-signed certificate in certificate chain**.
3. Reviewing the **subject** and **server_name** fields to determine the likely organization or website that controls the server.
4. Filtering results where there are legitimate self-signed certificates in use, such as in communications between IOT devices and the supporting cloud infrastructure.
5. Investigating the **id.resp_h** IP address to see what Autonomous System the session belongs to and whether it's a reasonable AS organization (such as the organization that matches the information on the server, or a commonly-used cloud hosting provider)
6. For remaining connections, use the values in the **cert_chain_fuids** to pivot to the certificates in the **x509** log and review the certificate details.

Focus your investigations by inspecting the local root certificate authority on the endpoint.

SELF SIGNED TLS SSL CERTIFICATE (OVERVIEW QUERY)

```
(#path="ssl" or #path=ssl_red) validation_status="self signed certificate"
| groupby([server_name], function=[collect(subject, limit=15), collect(issuer, limit=15)])
```

SSL CONNECTIONS WITH NON-OK CERTIFICATE VALIDITY (OVERVIEW QUERY)

```
(#path=ssl or #path=ssl_red) validation_status!="ok" validation_status=*
| groupby([id.resp_h, id.resp_p], function=[collect(server_name, limit=10), collect(subject, limit=10), collect(issuer, limit=10), collect(ja3, limit=10), collect(ja3s, limit=10), collect(validation_status, limit=10), collect(cipher, limit=10), collect(version, limit=10)])
```

CREDENTIAL ACCESS (TA0006)

Credential access is when the adversary tries to steal authentication materials, typically account names and passwords.

Brute Force (T1110)

An adversary attempts to gain unauthorized access by systematically guessing a user's password using a repetitive or iterative mechanism. Sometimes a brute force attack originates from a list of known information, increasing the likelihood of success.

For example, an attacker attempting to guess the password of an Active Directory account likely results in many connections to a Domain Controller on the LDAP port (389 or 636). An attacker attempting to discover API URLs in an e-commerce system generates many more connections to the web server than other clients in a similar time period and creates more HTTP status codes in the 400 and 500 range (errors) compared to other clients.

To look for a brute force attack:

1. In the conn log, aggregate by **id.orig_h**, **id.resp_h**, **id.resp_p**, **proto**, and (optionally) **service**.
2. Add a count for the number of operations and sort by the highest counts.
3. Choose a time period that makes sense, based on the size of the network/data set, starting small and gradually increasing.
4. Filter records that are obviously permissible, such as repeated contacts from network or application performance monitoring systems, vulnerability management systems, or business applications
5. For unknown or suspicious records, perform a deeper investigation on that behavior. For example, look for other connections originating from the remote IP address.
6. For protocols that can maintain connections over multiple transactions or attempts, look for long-standing connections. These long-standing connections can also indicate repetitive behavior.

Corelight Sensors include a script that logs connections that are maintained for longer than a set of thresholds, starting at ten minutes and continuing up to three days. If you are not a Corelight customer, but use open source Zeek, this script is available through the Corelight [GitHub page](#).

To hunt for long connections with the Long Connections package installed:

1. Examine the **notice** log.
2. Review the entries where the **note** is **LongConnection::found**,
3. Review each set of **id.orig_h**, **id.resp_h**, **id.resp_p** to understand whether these devices should have long connections.

To hunt for long connections without the Long Connections package installed:

1. Examine the **conn** log.
2. Gather a list of all connections with the following fields for each: **id.orig_h**, **id.resp_h**, **id.resp_p**, **proto**, **service**, and **duration** fields. This only includes connections that completed, either properly or via timeout. Currently-open connections are not represented in the results.
3. Sort the results by duration, bringing the longest connections to the top.
4. Investigate each result to determine whether it is legitimate or expected behavior.
5. Filter out expected behaviors and thoroughly investigate anything that seems suspicious.

Corelight also gives you the Encrypted Traffic Collection (ETC), which automatically looks for brute force password guessing attempts against SSH servers within a single connection.

POTENTIAL FORCED EXTERNAL OUTBOUND KERBEROS

```
(#path="conn" or #path="conn_red") history="Sh*" service="*krb*" local_orig="true"
local_resp="false"
| groupby([id.resp_h, id.resp_p, resp_cc], function=[collect(id.orig_h, limit=15),
collect(conn_state, limit=10), collect(service, limit=10), count(id.orig_h, distinct=true,
as="#_clients")])
```

POSSIBLE KERBEROS BRUTE FORCE ATTEMPT

```
#path="kerberos" client="*" | groupby(field=id.resp_h, function=count(field=id.orig_h, distinct=true, as=val)) | val > 100
```

MULTIPLE SSH BRUTE INFERENCE FROM SINGLE IP

```
#path="ssh" | split(inferences) | inferences="*BF*"  
| groupby(field=id.orig_h, function=count(field=id.resp_h, distinct=true, as=val)) | val > 3
```

Forced Authentication (T1187)

Some protocols automatically authenticate when a user accesses a resource without first checking to see if the resource being accessed is trusted. For example, an attacker can embed a reference in a Microsoft Office document to a file that's hosted on an attacker-controlled UNC path (`\servername\sharename\path\to\file`). When the user opens the file, the machine attempts to access the resource. The attacker-controlled server then challenges the machine for authentication, and under most circumstances, the victim machine automatically provides cached credentials, usually in the form of an NTLM hash. The attacker can then attempt to use the credentials for unauthorized access, usually through reversing the hash to get the password, or re-using the hash in a pass-the-hash attack.

This method requires the attacker to control server infrastructure. As a result, the most likely attack vector is spearphishing. The attacker phishing a user on the network, and the victim machine then reaches out to the attacker-controlled server across the internet. To hunt for this behavior, look for authentication across the internet:

1. Look in the `ntlm` log for any signs of NTLM authentication in which the destination IP is on the external network.
2. Look for entries in the `conn` log in which the service field contains `smb` (and/or `ntlm`), and `local_resp` is `false`.

In LLMNR or NBT-NS poisoning, an attacker listens to local LLMNR or NBT-NS broadcasts asking for a particular resource by name. The attacker then responds to the querying client spoofing the actual resource. If the resource is one that usually requires authentication, then the attacker can challenge the client for authentication. When the client authenticates, usually with a password hash, the attacker uses the credentials to impersonate the client and access resources.

You can effectively hunt for these attacks with Corelight data, but the sensor needs to be inside of the broadcast domain because broadcast traffic doesn't typically traverse routers. Typically you need to span or mirror entire VLANs, or forward LLMNR or NBT-NS traffic from client subnets and VLANs, to places on the network that Corelight is monitoring.

Look for `dns` logs where `id.resp_p=5355` (LLMNR) or `id.resp_p=137` (NBT-NS), and filter for records where the `answers` field is non-empty. Then count the number of distinct `query` fields per `id.resp_h`. This search yields IPs that respond to more than one name.

POTENTIAL WEBDAV FORCED AUTHENTICATION

```
definetable({{#path=conn or #path=conn_red) local_orig=true local_resp=false},  
name=outbound_conns, include=[id.orig_h, id.resp_h])  
| (#path="http" or #path="http_red") user_agent=/webdav/i  
| match(file=outbound_conns, field=[id.orig_h, id.resp_h])  
| groupby([id.orig_h, id.resp_h, id.resp_p], function=[collect(host, limit=10),  
collect(method, limit=10), collect(status_code, limit=10), collect(user_agent, limit=10)])
```

POTENTIAL FORCED EXTERNAL OUTBOUND NTLM

```
(#path="conn" or #path="conn_red") history="Sh*" service="*ntlm*" local_orig="true"  
local_resp="false"  
| groupby([id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(conn_state,  
limit=10), collect(service, limit=10)])
```

POTENTIAL FORCED EXTERNAL OUTBOUND SMB

```
(#path="conn" or #path="conn_red") history="Sh*" service="*smb*" local_orig="true"  
local_resp="false"  
| groupby([id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(conn_state,  
limit=10), collect(service, limit=10)])
```

POTENTIAL FORCED EXTERNAL OUTBOUND DCE_RPC

```
(#path="conn" or #path="conn_red") history="Sh*" service="*dce*" local_orig="true"  
local_resp="false"  
| groupby([id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(conn_state,  
limit=10), collect(service, limit=10)])
```

POTENTIAL FORCED EXTERNAL OUTBOUND GSSAPI

```
(#path="conn" or #path="conn_red") history="Sh*" service="*gssapi*" local_orig="true"  
local_resp="false"  
| groupby([id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(conn_state,  
limit=10), collect(service, limit=10)])
```

POTENTIAL FORCED LLMNR LOOKUP

```
definetable({{(#path=conn or #path=conn_red) local_orig=true local_resp=false},  
name=outbound_conns, include=[id.orig_h, id.resp_h])  
| (#path="dns" or #path="dns_red") id.resp_p="5355" answers[0]="/" query="*"  
| match(file=outbound_conns, field=[id.orig_h, id.resp_h])  
| groupby(field=[id.orig_h], function=[collect(id.resp_h, limit=15), collect(id.resp_p,  
limit=15), collect(query, limit=15), collect(service, limit=15), count(field=query,  
distinct=true, as=val)])  
| val > 2
```

POTENTIAL FORCED NETBIOS DNS LOOKUP

```
definetable({{(#path=conn or #path=conn_red) local_orig=true local_resp=false},  
name=outbound_conns, include=[id.orig_h, id.resp_h])  
| (#path="dns" or #path="dns_red")  
| split(answers)  
| (id.resp_p="137" or id.resp_p="138") answers="/"  
| match(file=outbound_conns, field=[id.orig_h, id.resp_h])  
| groupby(field=[id.resp_h, id.resp_p], function=[collect(id.orig_h, limit=10), collect(query,  
limit=10), count(field=query, distinct=true, as="#_query")])  
| "#_query" > 2
```

Network Sniffing (T1040)

You can't detect an intruder who is sniffing traffic on your network using network logs because the action is invisible; however, *you can detect an intruder by sniffing on your own network* because your adversary can't see it.

Corelight Sensors enable you to deploy an out-of-band sensor grid that generates linked logs. These logs speed reliable observation and detection, and assist in avoiding the pitfall of prevention dependence, while providing context for a deeper and more accurate historical analysis. As Rob Joyce, Chief of the NSA Tailored Access Operations division, put it in his 2016 USENIX talk, "We are doing nation-state exploitation...what can you do to defend yourself to make my life hard?"

DISCOVERY (TA0007)

The adversary is trying to learn about your environment.

Network Service Discovery (T1046)

To determine which devices on a network are exploitable, and the services available on those devices, an intruder can employ active scanning. Active scanning methods include:

- Horizontal scanning: Sending connection requests to a specific port across many IPs to see which IPs respond. For example, scanning across many devices on port TCP/22 typically reveals devices running an SSH server. Scanning across many devices on port TCP/445 can effectively enumerate Windows infrastructure.
- Vertical scanning: Sending connection requests to a single IP address across many ports to see which ports respond. This method lets attackers infer services available from that IP address.

Each of these methods can be performed using a free or commercially-available vulnerability scanner. These products often add other logic to check service availability, version information, and if services are vulnerable to known exploitation techniques.

If an intruder uses one or more of the above methods to attempt service discovery, the byproduct is a failed or rejected connection. In Corelight data, these are recorded in the **conn** log as connections with a **conn_state** of **S0** or **REJ** (initiated, and rejected), and typically have a **history** field where there is no **D** (post-syn data from the initiator). To look for network service scanning internal to the network:

1. Search for entries in the conn log where **conn_state** is **S0** or **REJ**.
2. Filter for records where **local_orig=true** and **local_resp=true**.
3. Group and count the results by the **id.orig_h**, and the number of unique **id.resp_p**, to assess the horizontal/verticalness of the scan.
4. Inspect the list, starting with the records that have the highest count of **id.resp_h** or **id.resp_p**.
5. Identify the originator (**id.orig_h**) and review the list of responders (**id.resp_h**) and ports (**id.resp_p**).
6. Determine whether the behavior is acceptable based on the identity of the source, the ports involved, and the destinations.

Not all items on the list are malicious. DHCP servers, for example, are commonly configured to ping an IP address to confirm if the address is in use before assigning it from the pool. Print servers with a large number of print queues attempt SNMP and/or network printing services to printers, even if those printers are offline. For this reason, print servers can cause large numbers of **S0** connections. Of course, software that scans legitimately, such as a corporate-sanctioned vulnerability scanner or an inventory management system, might appear in the list. Finally, network engineers conduct ad-hoc network scanning for troubleshooting purposes. If you run across network scanning, modify the original query to omit the records that are known to be benign, then resume hunting.

NETWORK SERVICE SCANNING MULTIPLE IPS FOR OPEN PORT

```
(#path="conn" or #path="conn_red") (conn_state="S0" or conn_state="REJ") local_orig="true"
local_resp="true" !(history="*d*")
| groupby(field=id.orig_h, function=[count(field=id.resp_p, distinct=true, as="#_ports"),
count(id.resp_h, distinct=true, as="#_dest")])
| "#_ports" > 10
```

NETWORK SERVICE SCANNING MULTIPLE IPS

```
(#path="conn" or #path="conn_red") conn_state="S0" or conn_state="REJ"
| local_orig="true" local_resp="true" !(history="*d*")
| groupby(field=id.orig_h, function=[count(field=id.resp_h, distinct=true, as="#_dest"),
count(field=id.resp_p, distinct=true, as="#_port")])
| "#_dest" > 25
```

DOMAIN USER ENUMERATION NETWORK RECON 01

```
#path="dce_rpc"
| in(operation, values=["LsarLookupNames3", "LsarLookupSids3", "SamrGetGroupsForUser",
"SamrLookupIdsInDomain", "SamrLookupNamesInDomain", "SamrQuerySecurityObject",
"SamrQueryInformationGroup"])
| groupby(field=id.orig_h, function=[count(field=operation, distinct=true, as="#_ops"),
count(id.resp_h, distinct=true, as="#_dest"), collect(endpoint, limit=10), collect(operation,
limit=10), collect(named_pipe, limit=10)])
| "#_ops" > 6
```

MULTIPLE REMOTE SMB CONNECTIONS FROM SINGLE CLIENT

```
#path="smb_mapping" path="*"
| groupby(field=id.orig_h, function=[count(field=path, distinct=true, as="#_shares"),
collect(id.resp_h, limit=20), collect(path, limit=20)])
| "#_shares" > 20
```

Network Share Discovery (T1135)

The most common network sharing protocol abused by attackers is SMB, the standard for Windows file sharing. SMB is supported by every modern operating system. High-value documents that store PII, trade secrets, network diagrams, and other sensitive data, typically live on SMB shares in enterprises of all sizes.

Scanning for and discovering shares on an SMB server is typically done using a DCE/RPC command on TCP port 445. Specifically, a connection to the "srvsvc" pipe—which shows up in the **dce_rpc** logs as an endpoint by the same name—is followed by a call to the **NetShareEnumAll** or **NetShareEnum** functions (called an **operation** in the log). These function calls are used for legitimate file-sharing purposes, and taken alone they are insufficient indicators of malicious intent. However, in combination with other indicators of lateral movement, they illustrate how an attacker moved laterally within a network. Prime targets for further investigation are ones that generate a large number of DCE_RPC function calls across a large number of hosts in a short period.

NETWORK SHARE DISCOVERY

```
#path="dce_rpc" (endpoint="srvsvc" or operation="NetshareEnumAll" or
operation="NetShareEnum")
| groupby(field=id.orig_h, function=[count(field=id.resp_h, distinct=true, as="#_dest"),
collect(id.resp_h, limit=10), collect(operation, limit=10), collect(endpoint, limit=10),
collect(named_pipe, limit=10)])
| "#_dest" > 3
```

Network Sniffing (T1040)

See [Credential Access: Network Sniffing](#)

Remote System Discovery (T1018)

The same principles for detecting [Network Service Scanning](#) apply to detecting Remote System Discovery. See this section for more information.

LATERAL MOVEMENT (TA0008)

Lateral movement is what adversaries use to enter and control remote systems on a network.

Remote Services: Remote Desktop Protocol (T1021.001)

The Microsoft Remote Desktop Protocol (RDP) is used to remotely control a Windows endpoint. This protocol can be abused by an attacker to gain unauthorized access to your network (see [Initial Access: External Remote Services](#)). Once an intruder is inside, they can use RDP to move laterally among devices.

RDP is one of the many protocols parsed by Corelight. For some environments, the presence of RDP, or its presence on specific systems, is sufficient to trigger an investigation. For networks where RDP is permitted, the Corelight **rdp** log is rich in information that helps establish whether a connection is legitimate, for example, recording data like keyboard layout, encryption levels, or client name for a connection.

When hunting with the **rdp** log:

1. Focus on the **id.orig_h**, **id.resp_h**, **id.resp_p**, and **cookie** fields. The **cookie** field can contain any arbitrary value sent by the RDP client to the server, but it often contains the username sent by the RDP client.
2. Aggregate the records based on these four fields and show a count for each unique set.
3. Iterate through the set and identify the origin and destination of each connection (e.g., you can use the records from the **dns** and **dhcp** logs).
4. Some RDP connections will use a non-standard keyboard layout. To look for this, examine the **keyboard_layout** field. Count the number of instances of each value and apply data stacking to look for outlying or rarely occurring values.
5. Identify the origin and destination and determine whether the non-standard keyboard layout is expected, for instance, if the origin user is known to have a non-English language as their primary language, and that language is the requested language in the RDP connection.

After you have this information ask several questions:

- Does the **cookie** value match the expected user at the source or destination machine?
- Is there a legitimate reason for the originator to be using RDP?
- Are there any users using RDP where you wouldn't expect that for their job function?

RDP DASHBOARD (OVERVIEW QUERY)

```
#path=rdp id.orig_h=*
| groupby([id.orig_h, id.resp_h, id.resp_p], function=[collect(cookie, limit=10),
collect(keyboard_layout, limit=10)])
```

RDP SUSPICIOUS KEYBOARD LAYOUT

```
#path="rdp"
| in(keyboard_layout, values=["*Arabic*", "Armenian - Armenia", "Farsi", "Pashto", "Swahili",
"Syriac", "Chinese *", "Mongolian (Mongolian)", "Mongolian *", "Tibetan *", "Uighur - China",
"Assamese", "Kannada", "* India", "Sanskrit", "Telugu", "Korean", "* Nigeria", "Wolof",
"Yoruba", "Catalan", "Rhaeto-Romanic", "Albanian - Albania", "*Cyrillic*", "FYRO Macedonian",
"Russian *", "Sorbian", "Uzbek (Cyrillic)", "Uzbek (Latin)", "Yakut", "Serbian", "Slovak",
"Slovenian", "Vietnamese"])
| groupby([id.orig_h], function=[collect(id.resp_h, limit=10), collect(id.resp_p, limit=10),
collect(cookie, limit=10), collect(keyboard_layout, limit=10)])
```

RDP POSSIBLE NON USER LOGIN, ABNORMAL SCREEN RESOLUTION

```
#path="rdp" destop_height<600 desktop_width<600
| groupby([id.orig_h], function=[collect(id.resp_h, limit=10), collect(id.resp_p,
limit=10), collect([desktop_width, desktop_height], limit=10), collect(cookie, limit=10),
collect(keyboard_layout, limit=10)])
```

Exploitation of Remote Services (T1210)

Exploitation of a software vulnerability occurs when an adversary takes advantage of a programming error to execute adversary-controlled code. This exploitation can happen in a program, service, or within the operating system software or kernel itself. A common goal for post-compromise exploitation of remote services is for lateral movement.

Given the complexity of today's enterprise networks, a variety of third-party and external services are often in use. These services allow attackers to gain initial access or to move laterally. All connections are logged within the **conn** log, however, more details may be available within protocol-specific logs depending on the nature of the remote service under attack. For example, you can monitor the **http** log file for suspicious and unexpected HTTP requests (such as OPTIONS requests).

```
_path: http
uid: CEeVS92Ljnr9jbW2J5
id.orig_h: 54.235.163.229
id.orig_p: 41855
id.resp_h: 192.168.0.2
id.resp_p: 80
trans_depth: 1
method: OPTIONS
host: host-90-236-3-35.mobileonline.telia.com
uri: *
version:1.1
```

Additionally, Corelight extracts information about software observed on the network into the **software** log. This file provides defenders valuable data to monitor for unexpected or unauthorized servers, vulnerable or out-of-date services, and unpatched client software.

```
path: software
host: 192.168.0.53
software_type: SMTP::MAIL_CLIENT
name: Microsoft Outlook Express
version.major: 6
version.minor: 0
version.minor2: 2900
version.minor3: 5512
unparsed_version: Microsoft Outlook Express 6.00.2900.5512
```

MULTIPLE WINDOWS ADMIN SHARE CONNECTIONS

```
#path="smb_mapping" path="*ADMIN$*"
| groupby(field=id.orig_h, function=[collect([id.resp_h, path], limit=10),
count(field=id.resp_h, distinct=true, as="#_shares")])
| "#_shares" > 3
```

WINDOWS SYSVOL FILE MODIFICATION

```
#path="smb_files" !(action="SMB::FILE_OPEN") path=/sysvol/i
| groupby([id.orig_h], function=[collect([id.resp_h, id.resp_p, action, path], limit=10)])
```

MULTIPLE KERBEROS TICKETS USED FROM SINGLE CLIENT

```
#path="kerberos" success="true" request_type="TGS" client="*" till="*"
| groupby(field=id.orig_h, function=[collect([client, service, till], limit=20),
count(field=client, distinct=true, as="#_clients")])
| "#_clients" > 5
```

MULTIPLE WINDOWS REMOTE REGISTRY SERVICE CONNECTIONS

```
#path="dce_rpc" endpoint="winreg"
| groupby(field=id.orig_h, function=[collect([id.resp_h, endpoint], limit=20),
count(field=id.resp_h, distinct=true, as="#_dest")])
| "#_dest" > 10
```

Remote Services: SMB/Windows Admin Shares (T1021.002)

Windows systems have hidden network shares that are accessible only to administrators and provide the ability for remote file copy and other administrative functions. Example network shares include C\$, ADMIN\$, and IPC\$.

Attackers often use SMB to connect to administrative shares on Microsoft Windows workstations and servers. They may want to learn more about the target, extract sensitive files, upload malicious payloads, or authenticate so that further tools and attacks can proceed.

Corelight monitors SMB traffic, including authentication attempts, allowing defenders to log and notice patterns of administrative authentication attempts as well as monitor SMB traffic to extract transferred files. The example demonstrates the action **FILE_OPEN** being performed using the hidden admin share. Corelight logs the action performed including Open/Rename/Delete/Write.

```
path: smb_files
uid: CB3Ezw2X3tYKtxunq
id.orig_h: 10.10.199.101
id.orig_p: 49710
id.resp_h: 10.10.199.31
id.resp_p: 445
action: SMB::FILE_OPEN
path: \\\10.10.199.31\admin$ 
name: <share_root>
size: 24576
times.modified: 2020-04-07T21:17:30.244159Z
times.accessed: 2020-04-07T21:17:30.244159Z
times.created: 2016-07-16T06:04:24.770745Z
times.changed: 2020-04-07T21:17:30.244159Z
```

ADMINISTRATIVE SHARE FILE CREATION

```
#path="smb_files" action="SMB::FILE_WRITE"
| lower(path, as=share)
| (share="*admin$" or share="*print$" or share="*fax$*" or share=/.*\[a-z]\$\$/)
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, action, share, name], limit=15), count()])
```

PSEXEC OVER SMB DETECTED

```
#path="smb_files"
| lower(field="name", as="name")
| name="*psexesvc*"
| groupby([id.orig_h], function=[collect([id.resp_h, id.resp_p, action, path, name], limit=15)])
```

SCHEDULE TASK ACCESS OR MANIPULATION OVER SMB

```
#path="smb_files" name="*ScheduledTasks.xml" path="*\\*\\SYSVOL*" !(action="SMB::FILE_OPEN")
| groupby([id.orig_h], function=collect([id.resp_h, id.resp_p, action, path, name], limit=15))
```

SMB SINGLE FILE CREATED THEN DELETED SUCCESSIVELY

```
#path="smb_files" name="*" !(name=<share_root>)
| groupby(field=name, function=[collect([id.orig_h, id.resp_h, id.resp_p, path, action]),
count(field=times.modified, distinct=true, as="#_times_modified")])
| "#_times_modified" > 5
```

REMOTE CREATION OF TEMP FILE IN SYSTEM32 FOLDER

```
#path="smb_files" name="*SYSTEM32\\*.tmp*" !(action="SMB::FILE_OPEN")
| groupby(id.orig_h, function=collect([id.resp_h, id.resp_p, action, path, name], limit=15))
```

COLLECTION (TA0009)

The adversary is trying to gather data to achieve their goal.

Archive Collected Data (T1560)

To conceal data, attackers may consolidate data into compressed archive files, such as Zip, RAR, TAR, or CAB files. To hunt for this obfuscation technique, use the files log.

To search for compressed files:

1. Search the **files** log, retrieving the **id.orig_h**, **id.resp_h**, **mime_type**, **total_bytes**, and **source** fields.
2. Remove records with uninteresting **mime_types** from the results, for example:
 - a. application/x-x509-*
 - b. application/ocsp*
 - c. image/*
 - d. audio/*
 - e. video/*
 - f. text/*
 - g. application/xml
 - h. application/chrome-ext
3. Continue until only compressed files remain

MULTIPLE COMPRESSED FILES TRANSFERRED OUTBOUND

```
(#path="files" or #path="files_red") !(total_bytes="0") local_orig=true
| in(mime_type, values=["application/vnd.ms-cab-compressed", "application/warc",
"application/x-7z-compressed", "application/x-ace", "application/x-arc",
"application/x-archive", "application/x-arj", "application/x-compress",
"application/x-cpio", "application/x-dmg", "application/x-gzip", "application/x-lha",
"application/x-eet", "application/x-lrzip", "application/x-lz4", "application/x-lzh",
"application/x-lzma", "application/x-lzip", "application/x-rar", "application/x-rpm",
"application/x-rpm", "application/x-stuffit", "application/x-tar", "application/x-xz",
"application/x-zoo", "application/zip"
])
| groupby(field=id.orig_h, function=[collect([id.resp_h, id.resp_p, mime_type, sha1],
limit=15), sum(total_bytes, as="sum_bytes"), count(field=sha1, distinct=true, as="#_files")])
| "#_files" > 25
```

MULTIPLE COMPRESSED FILES TRANSFERRED OVER HTTP

```
defineTable(query={({#path=conn or #path=conn_red local_orig=true local_resp=false}),
include=[id.orig_h, id.resp_h], name="outbound_conns")
| (#path="http" or #path="http_red") (method="POST" or method="PUT") !(referrer="*")
request_body_len>0
| split(orig_mime_types)
| in(orig_mime_types, values=["application/vnd.ms-cab-compressed", "application/warc",
"application/x-7z-compressed", "application/x-ace", "application/x-arc",
"application/x-archive", "application/x-arj", "application/x-compress",
"application/x-cpio", "application/x-dmg", "application/x-eet",
"application/x-gzip", "application/x-lha", "application/x-lrzip", "application/x-lz4",
"application/x-lzma", "application/x-lzh", "application/x-lzip", "application/x-rar",
"application/x-rpm", "application/x-stuffit", "application/x-tar", "application/x-xz",
"application/x-zoo", "application/zip"])
| match(file=outbound_conns, field=[id.orig_h, id.resp_h])
| groupby(field=id.orig_h, function=[collect([id.resp_h, id.resp_p, host, method, uri,
status_code, referrer, orig_mime_types], limit=15), count(field=uid, distinct=true,
as="#_conns")])
| "#_conns" > 25
```

POSSIBLE DATA COLLECTION OVER SMB

```
#path="smb_files" name="*" !(name("<share_root>"))
| groupby(field=id.orig_h, function=[collect([id.resp_h, id.resp_p, action, path, name],
limit=20),count(field=name, distinct=true, as="#_files")])
| #_files > 50
```

POSSIBLE DATA COLLECTION RELATED TO OFFICE DOCS AND EMAIL ARCHIVES AND PDFS

```
#path="smb_files" name="*" !(name("<share_root>"))
| in(name, values=["*.doc", "*.xls", "*.ost", "*.pst", "*.pst1", "*.pdf", "*.vss", "*.viz"])
| groupby(field=id.orig_h, function=[collect([id.resp_h, id.resp_p, action, path, name],
limit=20),count(field=name, distinct=true, as="#_files")])
| "#_files" > 30
```

Automated Collection (T1119)

Attackers can deploy automated tools on a compromised host to monitor intranet services for sensitive data and corporate secrets. These tools can include scripts to search for (and copy) information such as file type, location, or name at specific time intervals. Intruders may use remote access tools to conduct automated collection.

For example, a custom tool may query an intranet web server or an internal email server, polling regularly for new content. Corelight monitors multiple protocols including HTTP, email, MySQL, FTP, and SMB traffic to provide insight into these queries.

When hunting for automated collection use, defenders can identify automated tools by watching for repetitive queries or regularly scheduled connections. For example, if an intruder is web scraping, there will be a large number of connections from a finite number of IP addresses. Additionally, you can use the SMB logs (**smb_files** or **smb_mapping**) to identify anomalous traffic patterns of devices accessing internal Windows file shares.

Data From Network Shared Drive (T1039)

Network shared drives are a treasure trove of sensitive corporate documents. Most enterprise networks host shared network drives using SMB, but some may rely on FTP, HTTP, or even RDP. Corelight can monitor access to shared network drives when protocols like SMB, FTP, or HTTP are used. Remote control protocols, like RDP, are also parsed in protocol-specific logs. Anywhere Corelight sees this traffic, it is monitored and logged in the protocol-specific log.

The following example demonstrates the **ftp** log. Corelight logs the command and arguments.

```
path: ftp
uid: C0EeI73um1Aw3rr0ib
id.orig_h: 10.0.0.11
id.orig_p: 45831,
id.resp_h: 119.74.138.214
id.resp_p: 21,
user: 1
password: <hidden>
command: RETR
arg: ftp://119.74.138.214/doc.exe
reply_msg: Transfer OK
```

SENSITIVE FILE ACCESS OVER SMB SHARE

```
#path="smb_files"
| in(name, values=[ "*.rsa", "*.pem", "*.dsa", "*.dit", "*.ecdsa", "*.ocsp", "*.ed25519",
  "*.p12", "*.pfx", "*.kdbx", "*keychain", "*keystore", "*keyring", "*pass.txt", "*password.txt",
  "*passwords.txt", "*.bek", "*passwd", "*shadow", "*salesforce.js", "*psafe3",
  "*credentials.xml", "*localsettings.php", "*.mimi", "*.dmp", "*.dump", "*hiberfil.sys",
  "*1.txt", "*.kirbi", "*.ost", "*.pst", "*groups.xml", "*.bak", "*.ovpn", "*.sqlite", "*.sqlite3",
  "*sqldump"])
| groupby([id.orig_h], function=[collect([id.resp_h, id.resp_p, action, path, name],
  limit=20), count(name, distinct=true, as="#_files")])
```

SENSITIVE FILE ACCESS ON ADMIN NETWORK SHARE

```
#path="smb_files" path="*ADMIN$*"
| lowercase(name)
| in(name, values=[ "*\\\\\\mimidrv*", "*\\\\\\lsass*", "*\\\\\\windows\\\\\\minidump\\\\*",
  "*\\\\\\hiberfil*", "*\\\\\\sqldmpr*", "*\\\\\\sam*", "*\\\\\\ntds.dit*", "*\\\\\\security*"])
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, action, path, name], limit=20),
  count(name, distinct=true, as="#_files")])
```

COMMAND AND CONTROL (TA0011)

The adversary is trying to communicate with compromised systems to control them.

Data Obfuscation: Protocol or Service Impersonation (T1001.003), Non-Standard Port (T1571)

Adversaries may use a well known or alternative port associated with an existing protocol to avoid more detailed inspection.

Hunting for C2 channels over commonly used ports is difficult, but not impossible. To look for C2 channels, search for well-known ports that are being used with an uncommon service.

When hunting for C2 using commonly used ports:

1. Initially focus on the **service** field, and search the **conn** log for entries where the **service** field isn't what you would expect for the standard port (the service field could be null or an unexpected service).
 - a. Start with the most common protocols:
 - TCP:80 (HTTP) TCP:443 (HTTPS)
 - TCP:25 (SMTP)
 - TCP/UDP:53 (DNS)
2. Corelight's Encrypted Traffic Collection contains a package titled Encryption Detection. Encryption Detection generates a notice when cleartext traffic is observed on usually encrypted ports. Observing notices for **Viz::UnencryptedService** highlights this behavior and helps you identify potentially malicious connections using common ports.

The Corelight Encrypted Traffic Collection package also has a feature that notifies you when a session uses instant encryption. The package looks for pre-shared keys or encrypted connections that begin without a traditional key negotiation. Observing notices for **Viz::CustomCrypto** highlights this behavior and helps you identify potentially malicious connections using common ports.

Additionally, you can use the Corelight **dpd** and **weird** logs to identify unexpected protocol behavior. These logs show debugging and parsing errors and identify out-of-specification usage of common ports and protocols—which might indicate malicious activity or covert use of known ports and protocols.

```
_path: dpd
uid: C5LNtk1n9NkT8m300j
id.orig_h: 192.168.0.54
id.orig_p: 52841
id.resp_h: 54.89.42.30
id.resp_p: 80
proto: tcp
analyzer: HTTP
failure_reason: not a http request line
```

COMMON PORT WITH UNUSUAL SERVICE

```
(#path=conn or #path=conn_red) id.orig_p<1024 local_orig=true local_resp=false
| in(service, values=["*http*", "*ssl*", "*rdp*", "*ssh*"])
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, conn_state, service], limit=15),
sum(orig_bytes, as="sum_orig_bytes")])
```

Encrypted Channel (T1573)

See the Data Obfuscation: Protocol or Service Impersonation/Non-Standard Port section for a description of Corelight's Encryption Detection package, the `dpd` log, and the `weird` log. These help you identify potential custom cryptographic protocols.

SSH INFERENCE ABNORMAL CLIENT ACTIVITY

```
#path="ssh"
| split(inferences)
| in(inferences, values=["ABP", "BF", "BFS", "RSI", "RSL", "RSP"])
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, inferences, client]), count()])
```

CUSTOM CRYPTOGRAPHIC INFERENCE DETERMINED BY CORELIGHT

```
#path="notice" note="Viz::CustomCrypto"
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, note, msg, sub], limit=10),
count()])
```

Fallback Channels (T1008), Multi-Stage Channels (T1104)

Adversaries have been known to split communications between different protocols, using one for inbound C2 and another for outbound data. This allows for the communication to bypass firewall restrictions.

Malware that splits communication between two hosts for instructions and for exfiltration introduces a new challenge for defenders. Recognizing the linkage between suspicious control traffic and large data transfers is challenging, but Corelight provides packages and frameworks that synthesize data. For example, there is a package for determining the producer-consumer ratio for connections that identifies imbalanced, and possibly suspicious, data transfers. Additionally, the Intelligence Framework enables coordination with other defenders by identifying possible indicators of compromise (IP addresses, email addresses, and domain names) in Corelight data.

It's difficult to correlate attackers using different communication methods and channels but Corelight content, along with Zeek frameworks and packages can help. They allow defenders to identify the hidden channels discreetly, providing multiple opportunities for detection.

Beyond watching for the previously mentioned C2 communication mechanisms, here are some other signs available in Corelight data:

- Use the **conn** log to identify communication patterns that indicate additional channels (e.g., using **id.orig_h** and **id.resp_h** to narrow connections to a time window and observe connections between the hosts that include odd ports, failed or refused connections, or interesting/suspicious elements).
- Use Corelight (ETC), or self-developed content, in conjunction with connection log discovery to find potential relationships between overlapping, adjacent, or interesting connections.
- Search for sequences of connections to unrelated hosts using different protocols or events in the **dpd** and **weird** logs as described in Data Obfuscation: Protocol or Service Impersonation/Non-Standard Port.

Ingress Tool Transfer (T1105)

Intruders typically move files onto compromised systems—both tools that can assist with further lateral movement, and/or sensitive files designed for exfiltration. Those files will typically move over an HTTP(S), SSH, or SMB connection.

For files moving over plaintext HTTP, details like the remote host name and the name and MIME type of the file being transferred can be useful indicators; users should also consult the **files** log for the hashes of files being moved, as many popular attacker tools have known cryptographic hashes that make identifying them easy. In the case of HTTPS, defenders can use the IP address of the remote system, as well as the certificate details noted in the **ssl** log (i.e., organization name, FQDN of the remote host from the CN, etc.) to look for anomalous connections.

Intruders copy files from one endpoint to another as they move laterally among compromised assets. Traditionally, file copies to or from Unix/Linux systems occur over the SSH protocol using the **scp** command. For Windows systems, remote file uploads or downloads typically happen over SMB, but also may use SSH via PUTTY.

Corelight Sensors with the ETC SSH inferences package enabled extend the **ssh** log. The extension includes an **inferences** field that adds inferred characteristics about the SSH traffic. For example, if the session is being used to move files, or if it is interactive:

- **LFU:** Large File Upload
- **LFD:** Large File Download
- **KS:** Keystrokes

To begin hunting for interesting SSH sessions use the **inferences** field in the ETC SSH package:

1. Identify sessions where the **inferences** field contains **LFU**, **SFU**, **LFD**, or **SFD**
2. Determine whether file activity via SSH is legitimate and expected

Corelight Sensors are preloaded with the MITRE BZAR (Bro/Zeek ATT&CK-Based Analytics and Reporting) package. MITRE BZAR identifies MITRE ATT&CK techniques for remote file copy, namely files being copied to C\$ or ADMIN\$ Windows file shares. This package generates entries in the **notice** log, as depicted here.

```
path: notice
uid: CiAtaM363GcEbU63zk
id.orig_h: 192.168.38.104
id.orig_p: 65431
id.resp_h: 192.168.38.102
id.resp_p: 445
fuid: FSeaVF4qnjl8cT3HF8
file_mime_type: text/plain
file_desc: Windows\\Temp\\hbaVJpzdnG
proto: tcp
note: ATTACK::Lateral_Movement_Extracted_File
msg: Saved a copy of the file written to SMB admin file share
sub: 2020-10-23/6f24ac6ce591baf02acd64684f596d2db0ec97c0
src: 192.168.38.104
dst: 192.168.38.102
p: 445
```

Even if you do not enable the MITRE BZAR package on your Corelight Sensor, Corelight still logs SMB share access in the **smb_mapping** log and file access and modification in the **smb_files** log.

The logs below illustrate the data contained in the Corelight family of SMB logs:

```
_path: smb_mapping
uid: CiAtaM363GcEbU63zk
id.orig_h: 192.168.38.104
id.orig_p: 65431
id.resp_h: 192.168.38.102
id.resp_p":445
path: \\\192.168.38.102\c$
share_type: DISK
```

```
_path: smb_files
uid: CiAtaM363GcEbU63zk
id.orig_h: 192.168.38.104
id.orig_p: 65431,
id.resp_h: 192.168.38.102
id.resp_p: 445,
action: SMB::FILE_OPEN
path: \\\192.168.38.102\\C$
name: Windows\\Temp\\hbaVJpzdnG
size: 1894,
times.modified: 2019-12-31T10:28:02.800834Z
times.accessed: 2019-12-31T10:28:02.753959Z
times.created: 2019-12-31T10:28:02.566496Z
times.changed: 2019-12-31T10:28:02.800834Z"
```

To hunt for lateral movement:

1. Start by searching the **smb_files** logs, and focus on the **id.orig_h**, **id.resp_h**, **path**, and **name** fields
2. Filter out records where **id.resp_h** is a known file server, which reduces the results to potentially interesting connections
3. Review the **path** and **name** fields to identify which share the file was accessed from or written to, and determine if the behavior is suspicious.
4. For additional context about the remaining interesting records, you can pivot to the **files** log, using the UID to collect specific information about the file(s). For example, the MD5/SHA1/SHA256 hash(es) are automatically calculated and can be used to identify known malware in external systems, such as VirusTotal.
 - a. There are also other fields and possibly logs available (e.g., **pe** log) that can be used to rule out uninteresting records.

POTENTIALLY INTERESTING USER AGENT AND MIME TYPE COMBINATION

```
defineTable(query={#path=conn or #path=conn_red} local_orig=true local_resp=false,
include=[id.orig_h, id.resp_h], name="outbound_conns")
| #path="http" or #path="http_red" | split(resp_mime_types)
| in(resp_mime_types, values=["application/java-archive", "application/mshelp",
"application/chrome-ext", "application/x-object", "application/x-executable",
"application/x-sharedlib", "application/x-mach-o-executable", "application/x-dosexec",
"application/x-java-applet", "application/x-java-jnlp-file", "text/x-php", "text/x-perl",
"text/x-ruby", "text/x-python", "text/x-awk", "text/x-tcl", "text/x-lua",
"text/x-msdos-batch"])
| lowercase(user_agent)
| in(user_agent, values=["*certutil*", "*powershell*", "*microsoft*", "*python*",
"*libwww-perl*", "*go-http*", "*java/*", "*lua-resty-http*", "*winhttp*", "*vb project*",
"*ruby*"])
| match(file=outbound_conns, field=[id.orig_h, id.resp_h])
| groupby([id.orig_h], function=[collect([id.resp_h, id.resp_p, user_agent, method, host,
resp_mime_types, uri], limit=15)])
```

EXECUTABLE FROM WEBDAV

```
(#path="http" or #path="http_red") (user_agent="*WebDAV*" or uri="*webdav*")
| split(resp_mime_types)
| resp_mime_types="*dosexec*" or uri="*exe"
| groupby(id.orig_h, function=collect([id.resp_h, id.resp_p, user_agent, method, host,
resp_mime_types, uri]))
```

Non-Application Layer Protocol (T1095)

Attackers often make use of a pair of techniques for hiding inside of legitimate traffic: sending their communications over a custom protocol on a commonly allowed port like 80, 443, or 53, and embedding their messaging inside of the structure of legitimate but typically less-monitored protocols like ICMP.

For the use of custom protocols on standard ports, see the Data Obfuscation: Protocol or Service Impersonation section for a description of Corelight's Encryption Detection package, the **dpd** log, and the **weird** log. These help you identify custom C2 communications that use non-standard encryption or violate traditional protocol specifications.

Malware sometimes employs standardized lower-level protocols like ICMP, UDP, and SOCKS to avoid detection as these protocols are rarely monitored. For example, malware authors might embed C2 instructions in an ICMP Echo Request ("ping") packet.

Corelight monitors all connections regardless of protocol, and stores connection data within the **conn** log. C2 channels that employ custom UDP protocols or TCP-based SOCKS protocols (but no standard application layer protocols) have **conn** log entries with no identifiable **service** field. These fields and logs provide visibility into traffic flows across the network—even ICMP, UDP, and SOCKS. For ICMP sessions, Corelight data contains more than just the source and destination, for example; packet counts, bytes transferred, and size of ICMP data for both the sender and recipient.

With this data, you have the information needed to discover abnormally large or frequent ICMP communications that can be indicative of C2. The log shown is a sample of the **socks** log.

```
_path: socks
uid: C5u9ig4ACZvweN5my6
id.orig_h: 192.168.0.2
id.orig_p: 55951
id.resp_h: 192.168.0.1
id.resp_p: 1080
version: 5
user: bob
status: succeeded
request.host: 192.168.0.2
request_p: 22
bound.host: 192.168.0.1
bound_p: 55951
```

To hunt for an intruder using a standard non-application layer protocol to tunnel information:

1. Search the **conn** log for entries where the **service** field is blank, **local_orig** is **true**, and **local_resp** is **false**
2. Aggregate those results by **id.orig_h**, **id.resp_h**, **id.resp_p** and summarize by count
3. Filter out 'normal' entries
4. Investigate any remaining items, focusing on the line items with the greatest count first

Proxy (T1090)

While the use of proxies doesn't itself prove the presence of an intruder, intruders can use proxies to "launder" connections to obscure the communication from defenders. There are many methods to observe this, including traditional analysis of the underlying connection (signature, anomaly, behavioral) and statistical analysis of connection properties. Specifically identifying proxied connections is critical for beginning hunting or investigation.

If you see a value in the **proxied** field of Corelight's **http** log, that means an HTTP connection was proxied. The **http** log captures proxy details from the HTTP headers. Search for any records in the **http** log that have a non-empty **proxied** field.

- **host**: the domain name of the website
- **id.orig_h**: the IP address of the proxy or reverse proxy
- **id.resp_h**: the IP address of the web server
- **proxied**: identifies the proxy and the original IP address of the client

For example, a client at IP 219.90.98.8 initiated this HTTP request. The request was proxied via 172.16.1.30 to the web server at 172.16.2.95.

```
host: www.totallyfakedomain.com
id.orig_h: 172.16.1.30 //the proxy
id.orig_p: 53,828
id.resp_h: 172.16.2.95 //the web server
id.resp_p: 80
method: POST
post_body: dXNlcm5hbWU9cm9vdCZwYXNzd29yZD1tb25rZXk=
proxied: X-FORWARDED-FOR -> 219.90.98.8 //the real client
status_code: 200
status_msg: OK
uri: /xmlrpc.php
user_agent: Mozilla/4.0 (compatible; MSIE 6.0; Windows NT 5.0)
log: http
```

Using this example, identify the proxy and determine whether it's internal or external. If it's external, evaluate the session and gain context, using Corelight data to decide whether or not to block it. If the proxy is internal, determine whether it's a legitimate piece of IT infrastructure, or if it is a rogue proxy set up to circumvent policy—shadow IT.

Additionally, SOCKS is a commonly used proxy protocol that Corelight Sensors natively parse. When SOCKS is encountered, a **socks** log is generated and records details on users and protocols. This information can be used to ensure that connections aren't malicious and comply with policy. In the **socks** log, focus on these fields:

- **id.orig_h**: the client IP address
- **id.resp_h**: the proxy IP address
- **request**: the domain or IP the client is attempting to access
- **user**: if it is an authenticated connection, the user using the proxy

EXTERNAL PROXY DETECTED (OVERVIEW QUERY)

```
(#path="http" or #path="http_red") proxied[0]!=""
!(id.orig_h="10.*" or id.orig_h="192.168.*" or id.orig_h = /(^172\.\1[6-9]\..*)/
    or id.orig_h = /(^172\.\2[0-9]\..*)/ OR id.orig_h = /(^172\.\3[0-1]\..*)/
    or id.orig_h="127.*" or id.orig_h="169.254.*" or id.orig_h="::1"
    or id.orig_h="fe80::*" or id.orig_h="fc00::*")
| split(proxied)
| groupby(id.orig_h, function=[collect([proxied, id.resp_h, id.resp_p, user_agent, method,
host, uri], limit=10)])
```

Web Service (T1102)

Attackers may use a legitimate external web service to relay data to and/or from a compromised system to hide in the noise. While this tactic makes identification more difficult, Corelight data—especially the **http**, **ssl**, **conn**, and **x509** logs—helps you identify suspicious connections. Looking for IOCs including URI, hostname, or specific certificate details (like SNI or CN) is a good place to start. The following provides a few examples of certificate fields that might warrant an investigation:

```
path: x509
id: FfUGTX1VqS1qR30Jm7
certificate.version: 3
certificate.serial: 00
certificate.subject: emailAddress=obama@us.com,0=Obama inc.,L=Gaza City,ST=Gaza Strip,C=12,CN=http://usrep3.reimage.com
certificate.issuer: emailAddress=obama@us.com,0=Obama inc.,L=Gaza City,ST=Gaza Strip,C=12,CN=http://usrep3.reimage.com
certificate.not_valid_before: 2010-04-01T13:17:48.000000Z
certificate.not_valid_after: 2011-04-01T13:17:48.000000Z
certificate.key_alg: rsaEncryption
certificate.sig_alg: sha1WithRSAEncryption
certificate.key_type: rsa
certificate.key_length: 1024
certificate.exponent: 65537
```

MULTIPLE ABNORMAL NON CONFORMING HTTP REQUESTS

```
(#path=weird or #path=weird_red) name=bad_HTTP_request
| groupby(id.resp_h, function=[collect([id.orig_h, id.resp_p, name, notice, source], limit=10), count(id.orig_h, distinct=true, as="#_client")])
| #_client > 10
```

HTTP TRAFFIC WITH NO HTTP HOST SET OR USER AGENT SET

```
(#path="http" or #path="http_red") ((host=* user_agent!=*) or (user_agent=* host!=*))
| groupby([id.orig_h], function=[collect([id.resp_h, id.resp_p, user_agent, method, host, status_code, uri], limit=15)])
```

EXFILTRATION (TA0010)

Automated Exfiltration (T1020)

If an attacker is using an automated means of exfiltration, data artifacts are captured in the Corelight data.

To look for exfiltration in your network, you can use a Zeek package developed to calculate Producer/Consumer Ratio (PCR). PCR values indicate whether flows are consumptive (download) versus productive (upload). PCR values range from -1 (consumptive) to +1 (productive). To hunt for exfiltration using this package:

1. Install and enable a PCR package, or calculate it on-the-fly (example below).
2. Generate a table of **id.orig_h**, **id.resp_h**, **id.resp_p**, and **pcr** from the **conn** log.
3. Limit your results to records where **local_orig** is **true** and **local_resp** is **false** to focus on outbound connections.
4. Reduce the results by filtering out records where **pcr** <= 0.
5. For each host generating flows where **pcr** >= 0, consider whether that host is expected to transmit data outside the network.

Another option is to use a SIEM to calculate the PCR using the information available in the Corelight **conn** log. The following LogScale query creates a table organized by host that contains the originating and responding bytes and a PCR value.

```
#path=conn (orig_bytes>0 or resp_bytes>0) | groupby([id.orig_h, id.resp_h],  
function=[sum(orig_bytes, as="sum_orig_bytes"), sum(resp_bytes, as="sum_resp_bytes")]) |  
pcr:=(sum_orig_bytes-sum_resp_bytes)/(sum_orig_bytes+sum_resp_bytes)
```

Data Transfer Size Limits (T1030)

An attacker may attempt to transfer data or files by "chunking" them into smaller pieces, to avoid hard-coded data transfer limits or thresholds. We will present two methods to hunt for this technique.

The first method analyzes data leaving the network based on source and destination pairs:

1. Generate a table from the **conn** log including the **id.orig_h**, **id.resp_h**, **id.resp_p**, and **sum(orig_bytes)**.
2. Sort the results by the largest **sum(orig_bytes)**.
3. Examine each host and determine if there is a legitimate reason for uploads to that destination.

The second method analyzes the frequency, and sizes, of outbound transfers from each source:

1. Generate a table from the **conn** log including **id.orig_h**, **id.resp_h**, **id.resp_p**, and **count(orig_bytes)**.
2. Sort the results by the largest **count(orig_bytes)**.
3. Examine the results and determine the reason for all the connections with the same amount of data flowing from the source to the destination.

MULTIPLE FILES SENT OVER HTTP WITH ABNORMAL REQUESTS

```
(#path=http or #path=http_red) referrer!=* request_body_len>100000000
| split(orig_mime_types)
| in(orig_mime_types, values=["application/vnd.ms-cab-compressed", "application/warc",
"application/x-7z-compressed", "application/x-ace", "application/x-arc",
"application/x-archive", "application/x-arj", "application/x-compress",
"application/x-cpio", "application/x-dmg", "application/x-eet", "application/x-gzip",
"application/x-gzip", "application/x-lha", "application/x-lrzip", "application/x-lz4",
"application/x-lzma", "application/x-lzh", "application/x-lzip", "application/x-rar",
"application/x-rpm", "application/x-stuffit", "application/x-tar", "application/x-xz",
"application/x-zoo", "application/zip"])
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, method, host, orig_mime_types],
limit=15), sum(request_body_len, as="sum_request_bytes"), count(uid, distinct=true,
as="#_conns")])
| #_conns > 10
```

CLIENT SENDING LARGE AMOUNT OF DATA

```
(#path=conn or #path=conn_red) orig_bytes>1000000000 resp_bytes<100000000
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, service], limit=15),
sum(orig_bytes, as=sum_orig_bytes), sum(resp_bytes, as=sum_resp_bytes), count()])
```

CLIENT TRANSFERRING LARGE AMOUNT OF DATA OVER HTTP

```
(#path=http or #path=http_red) request_body_len>10000000
| groupby(id.orig_h, function=[collect([id.resp_h, id.resp_p, method, host, status_code,
uri, user_agent], limit=10), sum(request_body_len, as=sum_bytes), count()])
```

REFERENCES

1. <https://attack.mitre.org>
2. When used as an intel indicator IP is considered brittle, due to the ease with which adversaries can move to a new host or provider.
3. Not all versions of RDP assert the username in the cookie field. Some just assert nothing, or gibberish. In those instances, you would have to infer it from the NTLM or Kerberos log.
4. <https://www.wired.com/story/untold-story-2018-olympics-destroyer-cyberattack/>
5. Please visit <https://packages.zeek.org/> for additional information about Zeek packages

CORELIGHT OPEN NETWORK DETECTION & RESPONSE PLATFORM

Get a demo

Easily deployed and available in on-prem and SaaS-based formats, Corelight combines the power of open source and proprietary technologies to deliver a complete Open Network Detection & Response (NDR) Platform that includes intrusion detection (IDS), network security monitoring, and Smart PCAP solutions.

<https://corelight.com/products/demo>



Corelight provides security teams with network evidence so they can protect the world's most critical organizations and companies. Our Open Network Detection and Response Platform enhances visibility and analytics, leading to faster investigations and expanded threat hunting. Corelight's global customers include Fortune 500 companies, major government agencies, and large research universities. Based in San Francisco, Corelight is an open-core security company founded by the creators of Zeek®, the widely-used network security technology.

info@corelight.com | 888-547-9497

The Z and Design mark and the ZEEK mark are trademarks and/or registered trademarks of the International Computer Science Institute in the United States and certain other countries. The Licensed Marks are being used pursuant to a license agreement with the Institute.

All rights reserved. © Copyright 2026 Corelight, Inc.